Solution to Exercise 3.10.1

A group G of order n with prime factorization $n = \prod_{i=1}^{r} p_i^{j_i}$ is nilpotent if and only if it contains exactly one subgroup of order $p_i^{j_i}$ for each $i = 1, \ldots r$. It follows from Lemma 3.5.3 that the marks $m_i = m_{G/H_i}$ for the conjugacy classes of nilpotent subgroups H_i can be easily determined from the table of marks `t` of G. In GAP the command `IsNilpotentTom(t,i)` may be used to test whether or not H_i is nilpotent. A nilpotent group G is elementary if and only if a cyclic subgroup of G of maximal order has prime power index.

Using the table of marks of $G :=$ McL we compute up to conjugacy all cyclic subgroups, all p-subgroups and all nilpotent subgroups H_i of G, more precisely, thw rows and columns in the table of marks corresponding to these H_i.

```
gap> t := TableOfMarks( "McL" );;   n := Length( MarksTom(t) );;
gap> cycs :=  Filtered( [1..n], i -> IsCyclicTom( t, i ) );;
gap> psubs := Filtered( [1..n], i -> IsPrimePowerInt( OrdersTom(t)[i] ) );;
gap> nilps := Filtered( [1..n], i -> IsNilpotentTom( t, i ));;
gap> nilncp := Difference(nilps,cycs);; nilncp := Difference(nilncp,psubs);
[ 33, 51, 62 ]
```

There are just three conjugacy classes of nilpotent subgroups of G, which are neither cyclic nor p-groups.

```
gap> for k in nilncp do
>     cc := Filtered( [1..k] , j -> MatTom(t)[k][j] <> 0 and j in cycs );
>     ord := OrdersTom(t)[cc[Length(cc)]];
>     ind := OrdersTom(t)[k]/ord;
>     Print(k, " : ", ord, "  ", ind, ",      ");
>    od;
33 : 6  2,      51 : 6  3,      62 : 12  2,
```

We have printed for $k \in \{33, 51, 62\}$ the order and index of a maximal cyclic subgroup of H_k. So it turns out that they are elementary groups. Thus a subgroup of G is elementary if and only if it is nilpotent.

The maximal subgroups of G isomorphic to $\mathrm{U}_4(3)$ and $\mathrm{U}_3(5)$ are conjugate to H_m with $m \in \{372, 369\}$. For both values of m we list (using Lemma 3.5.3) all j such that H_j is nilpotent and not contained in a conjugate of H_m:

```
gap> maxlist := List( ["U4(3)","U3(5)"], x ->
>                Position( OrdersTom(t), Size(CharacterTable(x)) ) );
[ 372, 369 ]
gap> notcont := List( maxlist, m -> Difference( nilps,
>                    Filtered([ 1..m ], j ->  MatTom(t)[m][j] <> 0 ) ) );
[ [ 7, 25, 27, 36, 37, 71, 83, 200 ],
  [ 3, 9, 13, 14, 15, 17, 20, 21, 22, 24, 27, 35, 36, 37, 38, 39, 40, 41,
      42, 43, 44, 45, 46, 47, 51, 62, 72, 73, 74, 75, 76, 77, 78, 79, 80,
      81, 82, 83, 84, 85, 86, 87, 88, 89, 90, 91, 92, 139, 140, 141, 142,
      143, 144, 145, 157, 158, 159, 160, 161, 162, 163, 164, 165, 201, 243,
      244, 245, 304 ] ]
```

```
gap> int := Intersection( notcont[1], notcont[2] );
[ 27, 36, 37, 83 ]
gap> OrdersTom(t){int};
[ 11, 14, 15, 30 ]
```

Thus there are just four conjugacy classes of nilptent subgroups H_j which are not contained in a maximal subgroup isomorphic to $\mathrm{U}_4(3)$ or $\mathrm{U}_3(5)$. Their orders are 11, 14, 15, 30 and they are clearly all cyclic.